

Translating Mathematical Vernacular into Knowledge Repositories

Adam Grabowski¹ and Christoph Schwarzweller²

¹ Institute of Mathematics, University of Białystok
ul. Akademicka 2, 15-267 Białystok, Poland
adam@math.uwb.edu.pl

² Department of Computer Science, University of Gdańsk
ul. Wita Stwosza 57, 80-952 Gdańsk, Poland
schwarz@math.univ.gda.pl

Abstract. Defining functions is a major topic when building mathematical repositories. Though relatively easy in mathematical vernacular, function definitions rise a number of questions and problems in fully formal languages (see [Dav03]). This becomes even more important for repositories in which properties of the defined functions are not only stated, but also proved correct. In this paper we investigate function definitions in the Mizar system. Though most of them are straightforward and follow the intuition, we also found a number of examples differing from mathematical vernacular or where different solutions seem equally reasonable. Sometimes there even do not seem to exist solutions not somehow “ignoring mathematical vernacular”. So the question is: Should we seek for some kind of standard, that is a “formal mathematical vernacular”, or should we accept that different authors prefer different styles?

1 Introduction

Mathematical knowledge management aims at providing both tools and infrastructure supporting the organization, development, and teaching of mathematics on computers. To this end large repositories of mathematical knowledge are of major concern in order to provide a basis the potential user can rely on. Thereby, the acceptance of such repositories and hence of mathematical knowledge management systems heavily relies on the way mathematics is presented to the user; thus the closer to “everyday” mathematics the used language is, the more likely users of the system will be found.

The language actually used by mathematicians, however, is rather vague and imprecise: working mathematicians use what is called the “mathematical vernacular” [DeB87], [KN04], a language rather to communicate than to be completely formal. As stated by Davenport [Dav03] “It turns out to be remarkably hard to write ‘correct’ mathematics in the mathematical vernacular.” The reason is that the knowledge implicitly used in the vernacular must be made explicit for “correct” mathematics. The same holds for knowledge repositories, especially if

such a repository is to be connected with a theorem prover or checker and not just a collection of (textual) definitions and theorems. Here, for example, we do not have “obvious” special cases not worth mentioning.

On the other hand existing theorem provers and checkers provide languages successfully used to formalize and prove numerous advanced theorems. The languages to do so, however, are usually highly specialized and hard to understand from the viewpoint of working mathematicians. The reason here is a clash between what mathematicians and computers – that is computer scientists who design and implement theorem provers and checkers – consider comfortable. For theorem proving it might be reasonable to use languages “bizar” to a mathematician, as the goal is “simply” to find a (representation and) a computer proof for a specially chosen theorem.

In mathematical repositories the situation is somewhat different: here we look for general methods describing (and proving) theorems from different – if not all – areas of mathematics. So the question is: Should we develop mathematical knowledge management systems as closely as possible to the vernacular of working mathematicians in order to please them as potential users? Or should we include other language elements or slightly different definitions in case they are more convenient from the theorem proving point of view?

In this paper we try to give an answer to this question by inspecting the Mizar language and the Mizar Mathematical Library. We focus on definitions, in particular function definitions, which are often given partially or by case distinctions (see [Dav03]). This “impreciseness” is not further considered by mathematicians: theorems are stated without really worrying about the “easy special cases”. In mathematical repositories, however, this is not possible and therefore Mizar provides language constructs to cope with such situations. In addition we also present example situations which

1. do not strictly follow the “mathematical vernacular” and
2. provide a more elegant reuse in a repository.

The plan of the paper is as follows. After a brief review how functions can be defined in Mizar in the next section, we start with an investigation of the empty set and its elements in section 3. This “easy” example already indicates, that there exist different possibilities to realize mathematical vernacular in repositories. That this is no accident is shown in section 4 and 5 where a number of examples from different areas such as trigonometric functions and arithmetics are presented. Problems concerning more involved topics such as modularity of repositories and ambiguities are discussed in section 6. These observations imply that maintaining and revising of repositories will stay an important topic in the future. Section 7 discusses software built for the Mizar Mathematical Library to support this task.

2 Defining Functions in Mizar

Mathematical function definitions often cannot be defined uniformly on its domains; there are defined by case distinctions such as for example the signum

function or even partially by giving additional conditions for the arguments as in the case of inverse trigonometric functions. Of course one can introduce new domains on which such functions are then totally defined; this, however, seems to be rather artificial and in addition would lead to an inflation of domains not acceptable in a mathematical repository.

The Mizar systems basically provides two language constructs to cope with such situations: the `assume`-clause to express restrictions of arguments and the `if`-clause for defining case distinctions. In this section we give some introductory examples for using (and abusing) these constructs before we discuss its implications for mathematical knowledge repositories.

A standard example for restricting domains is the square root functions which is defined for non-negative real numbers only. The straightforward Mizar definition is as follows.

```
definition let a be real number;
  assume 0 <= a;
  func sqrt a -> real number means
    0 <= it & it^2 = a;
end;
```

Note, however, that this definition implies that for every application of `sqrt` a non-negative argument `a` is necessary, that is one has to show or state as an assumption that $0 \leq a$. Things become more puzzled when considering for example trigonometric functions: `tan a` is defined only if `cos a` is not zero, we thus get

```
definition let a be real number;
  assume not ex k being Integer st a = Pi / 2 + k * Pi;
  func tan a -> real number equals
    sin(a) / cos(a);
end;
```

and, given `a`, to get the value `tan a` the assumption is evident and has to be shown explicitly. The situation looks different when it comes to case distinctions using the `if`-clause. Though defining functions this way requires proving consistency – the cases need not be distinct, so one has to show that the corresponding values are not contradictory – most examples are straightforward and intuitive such as

```
definition let x,y be real number;
  func min(x,y) -> real number equals
    x if x <= y otherwise y;
end;
```

Proving theorems involving such functions is rather straightforward and fits to intuition. A prominent exception, though, is the inverse z^n of a complex number z , which is usually considered as a partial function, 0^n being undefined. In Mizar, however, we find that n is defined as a total function with 0^n being equal to 0.

```

definition let z be complex number;
  func z" -> complex number means :: XCMPLX_0:def 7
    z * it = 1 if z <> 0 otherwise it = 0;
end;

```

Note that in Mizar division a/b is defined as $a*b^{-1}$. This means that $/$ is a total function too, and in particular that we can prove $a/0 = 0$ for every complex number a . We will see in section 4 some more implications of this definition.

Of course it is easy to “abuse” these language constructs by introducing unnecessary assumptions, the probably most prominent example is using non-empty sets where this is not necessary. So the question is not only how to provide assumptions that can be reasonably used later, but also to avoid false assumptions in a repository. As we will see this can be sometimes achieved by ignoring the “mathematical vernacular”.

3 How Many Elements Has the Empty Subset?

To start the discussion we present in this section some issues of the empty set and its elements. Though rather trivial at first sight, this well illustrates the problems arising when moving from “imprecise” descriptions to “complete formal” ones. We will see that though the definition is almost trivial, using it in the environment of a mathematical repository – that is combining the definition with other notations from set theory – needs some care.

The empty set is the set which contains no elements. Thus it is straightforward to define something like

```

definition
  func {} -> set means :: XBOOLE_0:def 1
    not ex x being set st x in it;
end;

```

Though not exactly in the scope of defining functions, we like to mention the following problem here: In mathematical repositories definitions do not stand alone; they have to be considered in the context of other notations, here for example finite and infinite sets. Obviously, the empty set is finite. But in a repository that is not true in advance, it’s just obvious in the “mathematical vernacular”. Thus in principle it is possible to have objects such as

```

let X be infinite empty set;

```

Generally speaking that’s no harm, because such a phrase is meaningless: it includes a contradiction, hence everything stated (and proved) for such objects is of no use. On the other hand the acceptance of a repository in which this is possible is at least questionable. Thus such “contradictable” objects should be ruled out. In Mizar this is done by the use of existential cluster registrations: before using an object of type `infinite empty set` its existence has to be shown.

Now let's have a look at the elements of the empty set. In Mizar we find the definition of the type `Element of X`, where `X` is a set.³ It is “clear” that `x` is an `Element of X`, if `x ∈ X`. With this there are no problems if `X` is non-empty: There is an element `x` in `X`, so the type `Element of X` is non-empty. If `X` is empty, however, there is no `x ∈ X`. One possibility now is to define the type `Element of X` for non-empty sets `X` only, but this seems not to fit with mathematical vernacular. Therefore in Mizar the type `Element of {}` is defined to be the empty set:

```
definition let X be set;
  mode Element of X means :: SUBSET_1:def 2
    it in X if X is non empty
    otherwise it is empty;
end;
```

This, however, does not fit to mathematical vernacular either, because the empty set is not an element of the empty set; but has the advantage that the type `Element of X` is well-defined for arbitrary sets `X`. We mention that though the empty set `{}` is of type `Element of {}`, this does not imply that `{}` \in `{}` is provable in Mizar, that is `{}` is still the empty set. So we see that even a notion as “obvious” as the empty set calls for basic decisions when being formalized, that is when moving from mathematical vernacular into a mathematical repository.

4 Trigonometric Functions

In this section we consider the definition of trigonometric functions in Mizar. Interestingly, we can find different approaches, one following the intuition and another one using that the inverse of 0 is 0. First, the logarithm of real numbers `a` and `b` is defined using exponentiation, in Mizar defined as a functor `to_power` (see section 5.3). Here, the usual “problematic” values for `a` and `b` have been ruled out using an assumption:

```
definition let a,b be real number;
  assume that a > 0 & a <> 1 and b > 0;
  func log(a,b) -> real number means :: POWER:def 3
    a to_power it = b;
end;
```

Consequently, theorems have to take these values into account, because the equality `a to_power log(a,b) = b` is valid only if the assumptions about `a` and `b` are fulfilled. We thus find theorems of the following kind.

```
theorem :: POWER:61
  a>0 & a<>1 & b>0 & c>0 implies log(a,b) + log(a,c) = log(a,b*c);
```

³ We remind here that in Mizar types have to be non-empty.

This approach follows what Davenport called the “conditional equation approach” in [Dav03]. The advantage is, that it is close to text book mathematics (though there assumptions are often not stated thoroughly) and makes the necessary assumptions explicit. On the other hand long lists of assumptions both lower readability of theorems and require of course re-stating them when using such theorems in other proofs.

What can improve things a bit here is the technique using “default values” as presented in the definition of the inverse function ". Remembering that / is a total function (compare section 2), the tangent function for real numbers can be defined simply as follows.

```
definition let th be real number;
  func tan(th) -> real number :: SIN_COS4:def 1
    sin(th) / cos(th);
end;
```

which actually means that $\tan(\text{Pi}/2)$ is defined to be 0. Note that given th we can now get the value $\tan(\text{th})$ without proving $\text{th} \neq \text{Pi}/2$. This also implies that a number of theorems can be stated using no assumptions, so for example

```
theorem :: SIN_COS4:2
  tan(-th) = - tan(th);
```

This may seem irritating at first sight for a reader not familiar with the basic definitions of the repository; but has the advantage that this theorem can be used without further prerequisites to be shown. Of course not all theorems can be stated this way, because $z * z = 1$ holds only if $z \neq 0$. Here Mizar formalizations fall back to the conditional approach, so for example we find

```
theorem :: SIN_COS4:8
  cos(th) <> 0 implies sin(th) = cos(th) * tan(th);
```

5 Arithmetics and Related Issues

5.1 The Greatest Common Divisor

As the greatest common divisor $\text{GCD}(a, b)$ is the largest number dividing both integers a and b , according to our intuitions such number does not exist in case of $a = b = 0$. Indeed, a quick tour through mathematical services available via WWW confirms these convictions: Wolfram’s MathWorld’s⁴ definition of gcd takes only positive integer numbers as arguments; according to Wikipedia⁵ both should not be zero simultaneously, similarly is the PlanetMath’s⁶ opinion, but we can read in Wikipedia that “it is useful to define $\text{gcd}(0, 0) = 0$ ”.

In the Mizar library, there are two definitions of the greatest common divisor: hcf for natural numbers and gcd with integer arguments which uses the notion of hcf and the absolute value in its definiens.

⁴ <http://mathworld.wolfram.com>

⁵ <http://www.wikipedia.org>

⁶ <http://planetmath.org>

```

definition let k, n be Nat;
  func k hcf n -> Nat means :: NAT_1:def 5
    it divides k & it divides n &
    for m st m divides k & m divides n holds m divides it;
end;

```

Based on the above, we can easily prove that

$$a = 0 \wedge b = 0 \iff a \text{ hcf } b = 0$$

for all natural a and b , and similarly for integers. Furthermore, claiming such definition we keep the connectedness with commutative rings, we also obtain a lattice of naturals with `gcd` and `lcm` as binary operations to be both distributive and complete.

5.2 The Integer Division

When inspecting the integer division in the Pascal programming language, the FreePascal compiler returns 'division by zero' error both with `div` and with `mod`. Since the Mizar system itself is coded in Pascal (and as one can easily see, some Mizar language constructions have been influenced by this programming language), we could expect a similar behaviour for the functions `div` and `mod` in the MML.

Since both are defined usually (see Wolfram's MathWorld) as:

$$m \text{ div } n = \lfloor m/n \rfloor, \quad m \text{ mod } n = m - n \lfloor m/n \rfloor, \quad (1)$$

both share the restriction of $n \neq 0$ as usual in the literature.

This is not violation of intuitions, but the MML contains the following definitions, somewhat closer to Euclid's *Elements*:

```

definition let k, l be natural number;
  func k div l -> Nat means :: NAT_1:def 1
    ( ex t being Nat st k = l * it + t & t < l ) or it = 0 & l = 0;

  func k mod l -> Nat means :: NAT_1:def 2
    ( ex t being Nat st k = l * t + it & it < l ) or it = 0 & l = 0;
end;

```

The above definition is a variant of which we wrote earlier (something like the `if`-clause), but with a slightly different (but equivalent classically) formulation.⁷ There is an agreement in the MML that 0 is an element of \mathbf{N} (to have both functions natural-valued), but there isn't any within mathematics in general: MathWorld writes that "Unfortunately, 0 is sometimes also included to the list

⁷ The difference between `natural number` and `Nat` (with the latter expanding to `Element of NAT`) which has origins in various treatment of element of the empty set has to be recalled here. All `Nats` obtain the attribute `natural` automatically due to the conditional cluster mechanism.

of ‘natural’ numbers” (as Bourbaki and Halmos do), quoting Ribenboim’s as the opposition (“... whenever convenient, it may be assumed that $0 \in \mathbf{N}$ ”).

In [GKP94] they state that extending `mod` to omit the assumption of the division by zero is important, but they do not explain explicitly which one should be taken: the divided number or 0. Inspecting the book we discovered that if we accept the equation (1) as the new definitions of the integer division functions in the MML, we obtain $x = x \bmod 0$, and this is also claimed in [GKP94] more often.

There are contexts in which division by zero can be considered well-justified. For example, in the extended complex plane \mathbf{C}^* it is defined to be a quantity known as complex infinity. This definition expresses the fact that, for $z \neq 0$, $\lim_{w \rightarrow 0} z/w = \infty$ (i.e., complex infinity). However, even though the formal statement $1/0 = \infty$ is permitted in \mathbf{C}^* , this does not mean that $1 = 0 \cdot \infty$, so zero does not have a multiplicative inverse. On the other hand, although $\mathbf{R} \subseteq \mathbf{C}$, it is not clear which way to go with the extensions (since to the extended set of real numbers both $+\infty$ and $-\infty$ are added and this is the case of the MML).

As a good example of the other way of definition extending we can quote `min*` as an opposition to an ordinary `min` function.

```

definition let A be finite non empty real-membered set;
  redefine func min A means  :: SFMASTR3:def 1
  it in A & for k being real number st k in A holds it <= k;
end;

definition let A be set;
  func min* A -> Nat means  :: HENMODEL:def 1
  (it in A & for k st k in A holds it <= k) if
  A is non empty Subset of NAT
  otherwise it = 0;
end;

```

These two objects are defined completely independently, but the latter became apparent to be useful when proving the Gödel’s Completeness Theorem in Mizar. Theoretically, generalizing `min*` we can replace an original `min` to simplify the library a bit. Generalizing can be also interesting from a purely scientific point of view (as e.g., formalizing rough sets with tolerances as described in [GS04] or [Gra05] instead of equivalence relations). But usually the loci of a definition cannot be just generalized because the information contained in it may be necessary to give the proper meaning of an introduced object.

5.3 The Power Operator

The consequence of parallel introducing of similar notions (motivated by the need of having their definitions close to the literature) can be observed in the case of the definition of the power function, which is composed with the help of various power operators defined earlier in MML (`#R` is defined as the limit of sequence of rational powers with of a given real number – with the assumption of the positive base, `#Z` is a integer power, with arbitrary real base).

```

definition let a, b be real number;
  func a to_power b -> real number means :: POWER:def 2
    it = a #R b if a > 0,
    it = 0 if a = 0 & b > 0,
    it = 1 if a = 0 & b = 0,
    ex k st k = b & it = a #Z k if a < 0 & b is Integer;
end;

```

Any efforts to change this definition should be made carefully, because the article with this definition is referenced in 46 other MML items 1407 times. Similar data for the other power operators: 415 references in 39 articles.

Note that this definition is an example of a definition of a partial function (and keyword `otherwise` is not used there), e.g. according to this definition we still don't know which is the value of $(-1)^{-\frac{1}{2}}$, but it gets the type `real number`.

5.4 Polynomials

Consider polynomials as a last example. The head term – and hence the head coefficient – of a polynomial are usually defined for non zero polynomials only (see for example [BW93]. From a theorem proving point of view, however, it seems convenient to define a “head term” for the zero polynomial also as follows: The head term of the zero polynomial equals the smallest term with respect to the given order. This is can be seen as an extension of the head term functor found in the literature.

```

definition
  let n be Ordinal, T be connected TermOrder of n,
      L be non empty ZeroStr, p be Polynomial of n,L;
  func HT(p,T) -> Element of Bags n means :: TERMORD:def 6
    (Support p = {} & it = EmptyBag n) or
    (it in Support p &
     for b being bag of n st b in Support p holds b <= it,T);
end;

```

This allows to formulate theorems about head terms for arbitrary polynomials. As a consequence, when later reusing such theorems the user need not always bother that the actual polynomial is not equal 0 – just like mathematicians. For example, we get

```

theorem :: TERMORD:22
  for n being Ordinal, T being connected TermOrder of n,
      L being non trivial ZeroStr, p being Polynomial of n,L holds
    term(HM(p,T)) = HT(p,T) & coefficient(HM(p,T)) = HC(p,T);

```

if also `HC(p,T)` is defined appropriately, e.g. equals the zero element of the underlying coefficient domain.

6 Modularity and Ambiguity

6.1 Modularity of the Library

Although the fundamentals of set theory in Mizar are established rather hard, e.g. Axiom of Choice can be proved (and it is actually in [Ban90]), the user can also modify his/her preferences. As a perfect example in arithmetics of alephs we can cite the Generalized Continuum Hypothesis introduced by Josef Urban in [Urb01].

definition

```
pred GCH means  :: CARD_FIL:def 12
  for N being Aleph holds nextcard N = exp(2,N);
end;
```

theorem :: CARD_FIL:31

```
GCH implies ( M is inaccessible implies M is strongly_inaccessible );
```

where M is again of the type `Aleph`.

This trick may be used, e.g. to state the Brouwer Fix Point Theorem for disks on the real euclidean plane as an assumption to prove the famous Jordan Curve Theorem⁸. As the bright side of this approach to the development of the library we can point out the possibility of development of the authors' favorite parts of mathematics in which they are experts, instead of spending most time on bridging the gap between the current and the desired state of the formalization of the theory. This could attract more mathematicians and as we believe it is one of the vital aims of math-assistants and also of the MKM project. Also the research frontier could be so reached faster – which could make the machine codification of recent mathematics more egalitarian.

The modular maintenance of systems could be a solution for someone's wishes to have some meta-assumptions, but the care is advised (e.g., the Axiom of Determinacy contradicts the Axiom of Choice which is proven in the MML, so the earlier should not be accepted as such an assumption). Maybe something like the `requirements` directive with more human-friendly access and giving possibility of defining author's own modules of this type could be an attractive solution.⁹

Clearly, this can also have some impact on the knowledge exchange between different systems, according to the Sacerdoti Coen's advice in [Sac03]: "Make implicit information explicit". Note however that the logical system standing behind the Mizar system is fixed, and Mizar developers rather do not anticipate

⁸ Actually it is meaningless since Kornilowicz and Shidama proved this version of the Brouwer Fix Point Theorem in February 2005 as the `BROUWER` article accepted to the MML. The one-dimensional case is pretty old.

⁹ As of version 7.0.04 of the Mizar system, there are five modules of this type available: `BOOLE`, `SUBSET`, `HIDDEN`, `ARITHM`, `NUMERALS`, `REAL`, where first two introduce automatization of boolean operations on sets, the latter three – calculations on numbers. The detailed exposition of the topic is included in [Nau04].

change of this policy (e.g. from the classical into the constructive logic) in the future. Another drawback is that stating some significant and influential theorems without proofs we allow for a gray area of practically unverified and – hence void in some sense – mathematics. Machine verification does not tend to eliminate completely thorough peer-review system of checking mathematical papers.

6.2 Ambiguities

In a distributed knowledge repository it is hard to establish a high unification level (compare $1/0 = 0$, which is provable in HOL/Mizar, its negation is provable in IMPS, both the formula and its negation are not provable in Coq, or being not a correct formula in PVS), so there is a need to exchange information about the mathlore (as in QED Manifesto they wanted to call “knowledge that is neither taught in classes nor published in monographs”) which is accepted (or rather where it was rejected). As a mathlore we mean here not only basic facts which are commonly accepted, but also the formulation of definitions of basic notions.

But what to do with freshmen which are not well acquainted with mathlore? Anyone remembers from the school that the division by zero is not allowed as a rule, and noone complained, so why to think about pros to have some value for 0 divided by 0? The answer given in the MML is not obligatory in didactics: our recent teaching policy is not to use the whole repository, but to prepare small working environments building from scratch [Ret04], in which some other decision can be done. The reason is also that in this way it is independent from the Mizar library which evolves rapidly so the update during a semester could be hardly acceptable.

Safe choice is also not to decide by oneself and to keep different definitions and theorems in parallel. Thus, we can measure the use of notions by statistical means after some time and leave only the one which is most often used (as it can be done e.g. in case of `min` and `min*`).

7 Improving the Library

As mathematics assistance systems are designed as a tool offering machine help for human researchers, many of the decisions about chosen approach are taken on the user’s side depending on the various (subjective as a rule) criteria: elegance, faithfulness to mathematical standards, feasibility, etc.

However, especially if the cooperation between various systems is taken into account, much improvement of a repository can be done in a highly automatic way. The quality is to be measured by statistical, so quantitative means. As it is clear however, “short” does not mean “readable” and this is a serious drawback when thinking about reusability of proofs and their clarity for people. The de Bruijn factor, which is defined by Wiedijk as the quotient of a size of formal representation by its informal original can be a dead end sometimes.

All Mizar distributions contain the bunch of programs aiming at reviewing a Mizar article and which hence may lead to the enhancing of a human work

done by hand. The Library Committee of the Association of Mizar Users uses a collection of editing versions of the mentioned programs.

The software inspects a Mizar text focusing on three main activities:

- shortening and clarifying proofs;
- improving definitions’ and theorems’ level of generality;
- marking block and items which are just not used anymore.

There is a need for controlling of the necessity of some parts of the proofs written by human. As a rule, this can be done via stepwise refinement. There are few stages on which such control can be performed (and this is the case of the Mizar library):

Irrelevant premises

This is the most unproblematic and the most popular control which can be performed when writing a Mizar article. `relprem` reviews which references are not needed for the justification of a sentence.

Checking unused labels

Very often removed unused premises are just library references (for definitions and theorems proven in MML already), but sometimes the calling by a local fact is written accidentally. If any other sentence also does not use this labelled item, after the `chklab` pass such label is marked as unnecessary. Still though, the sentence can be needed in a proof via simple linking by the next one (the reserved word ‘`then`’ in such a case).

Inaccessible part of proofs

The program `inacc` points out sentences which are neither labelled nor linked (elements of a proof skeleton are not marked as erroneous).

Finding trivial proofs

Although Mizar proofs are hierarchical (in the sense considered e.g., by Lamport), sometimes after the aforementioned transformations nested proofs can be simplified by the program `trivdemo` to a simple justification, that is to a list of references preceded by the keyword ‘`by`’.

Irrelevant suppositions

As unnecessary assumptions (in the sense of elements of proof skeleton, not just as premises) are not marked by any of the programs mentioned before as vital element of proofs, this software operates on the stage of theorem formulation than proof transformation.

This program (`relsup`) is not freely available in the distribution. Explicit formulation of some assumptions in a proof may be forced by the so-called definitional expansions and hence not used directly. They are needed however and their automatic removal could result in an error in the proof skeleton and marking them as erroneous can be highly confusing, especially for an unexperienced author.

The above ordering of these programs reflects their preferred calling sequence.

The only controversial exception of the reviewing software is `relinfer` program (so it was excluded from our enumeration), which points out the unnecessary steps in a proof (and the references should be added to the next step).

It can exceptionally shorten proofs but it may result in poor readability of the text:

- some sentences which are important for the proof technically are marked as irrelevant steps, but their removal may force the user to repetition of the same library reference;
- the removal may be accidental in some sense, that is steps which are crucial for human understanding of the idea of a proof, but are still unnecessary for machine (e.g., unwinding definitions – definitional expansions). Here the tendencies to reduce the de Bruijn factor can be misleading.

We also have software which detects unused variable occurrences, irrelevant private predicates and functions, marks unnecessary type changing statements, etc.

Besides the aforementioned proof transformations which are performed very often, some other checkouts are done occasionally. There is a software which checks if there are equal theorems in the library, and what’s more interesting, if a theorem is a consequence of another (although due to the large library, both use a lot of resources). The latter one is often not very unlikely: to formulate statements as equivalences is the usual mathematical practice, very often though some assumptions are needed only for one of the implications.

We still do not have any automatic control if the definitions are repeated (authors would have like to introduce independently e.g. closure operators using different structures), so we can speak about the detection of ‘equal theorems’ rather than ‘equivalent’ ones. So the role of careful peer-reviewing of a repository is so important, especially if we take into account a large repository of mathematics, written by many authors, so rather heterogeneous in a style of writing. Quantitative parameters of the MML (some 40 thousand of theorems and lemmas, nearly 8 thousand definitions authored by more than 160 authors) justify the necessity of continuous revising of Mizar articles.

8 Conclusions

In this paper we have considered how the mathematical vernacular can be realized in mathematical repositories, thereby focusing on function definitions. The inspection of the Mizar Mathematical Library has shown that its authors used a number of different styles such as the “conditional” style using partial functions or the “extension” style as used in the definition of ". Sometimes even more than one definition is available. It seems to us that these different styles exist due to a clash between (a) working in a formal language close to the mathematical vernacular which (b) is also used to prove the theorems stated. Strictly following the mathematical vernacular sometimes leads to rather tedious formal proofs, so that some authors decide to modify their definitions in order to ease the proving task.

The problem becomes more evident in a repository with a large number of developers and users: here, of course, it is impossible to have an open system

without ending up with different realizations of the mathematical vernacular. Hence, should we seek for a kind of standard, that is a “formal mathematical vernacular”? Though we believe that this can be done in general, it seems hardly possible to fix all the details theorem proving introduces into our repositories. Allowing for different realizations, on the other hand, could of course lower the acceptance of users by confusing them. Also, extending or reusing developments by other authors gets more complicated in case the vernacular of the new author does not fit to the first author’s one.

What we can try to do is organize our repositories in such a way that both authors and potential users have the possibility to identify the basic decisions theories and developments rely on. As we have illustrated this also includes the definition of functions. This is not a trivial task, because as already mentioned the large number of Mizar authors has even led to duplication of definitions or theorems. Consequently, we always have to keep track of the development by permanently revising and cleaning up our repositories. The goal must be to automate this as far as possible. A step into this direction are the Mizar tools presented in section 7. They are, however, in most cases still working on the proof transformation level, so that their further development into a “more intelligent direction” is desirable. We should, however, always keep in mind – especially if we try to develop systems for working mathematicians – that if we break rules accepted widely by mathematicians, this has to be sufficiently justified.

References

- [Ban90] G. Bancerek, *Zermelo Theorem and Axiom of Choice*, Formalized Mathematics, 1(2), pp. 265–267, 1990.
- [BW93] T. Becker and V. Weispfenning, *Gröbner Bases – A Computational Approach to Commutative Algebra*; Springer Verlag, 1993.
- [DeB87] N.G. de Bruijn, *The Mathematical Vernacular, a language for mathematics with typed sets*, in P. Dybjer et al. (eds.), Proc. of the Workshop on Programming Languages, Marstrand, Sweden, 1987.
- [Dav03] J. H. Davenport, *MKM from book to computer: a case study*, in: A. Asperti, B. Buchberger, and J. Davenport (eds.), Proc. of MKM 2003, Lecture Notes in Computer Science 2594, pp. 17–29, 2003.
- [Gra05] A. Grabowski, *On the computer-assisted reasoning about rough sets*, in: B. Dunin-Kępcicz et al. (eds.), Monitoring, Security, and Rescue Techniques in Multiagent Systems, Advances in Soft Computing, Springer, pp. 215–226, 2005.
- [GS04] A. Grabowski and Ch. Schwarzweller, *Rough Concept Analysis – theory development in the Mizar system*, in: A. Asperti, G. Bancerek, and A. Trybulec (eds.), Proc. of MKM 2004, Lecture Notes in Computer Science 3119, pp. 130–144, 2004.
- [GKP94] R. E. Graham, D. E. Knuth, and O. Patashnik, *Concrete Mathematics*, Addison-Wesley, 1994.
- [KN04] F. Kamareddine and R. Nederpelt, *A refinement of de Bruijn’s formal language of mathematics*, Journal of Logic, Language and Information, 13(3), pp. 287–340, 2004.

- [Nau04] A. Naumowicz and Cz. Byliński, *Improving Mizar texts with properties and requirements*, in: A. Asperti, G. Bancerek, and A. Trybulec (eds.), Proc. of MKM 2004, Lecture Notes in Computer Science 3119, pp. 190–301, 2004.
- [Ret04] K. Retel and A. Zalewska, *Mizar as a tool for teaching mathematics*, in Proc. of Mizar 30 workshop, Białowieża, Poland, 2004, available at <http://www.macs.hw.ac.uk/~retel/papers/KRetelAZalewska.pdf>.
- [RT01] P. Rudnicki and A. Trybulec, *Mathematical Knowledge Management in Mizar*; in: B. Buchberger, O. Caprotti (eds.), Proc. of MKM 2001, Linz, Austria, 2001.
- [RT03] P. Rudnicki and A. Trybulec, *On the integrity of a repository of formalized mathematics*; in: A. Asperti, B. Buchberger, and J. Davenport (eds.), Proc. of MKM 2003, Lecture Notes in Computer Science 2594, pp. 162–174, 2003.
- [Sac03] C. Sacerdoti Coen, *From proof-assistants to distributed knowledge repositories: tips and pitfalls*, in: A. Asperti, B. Buchberger, and J. Davenport (eds.), Proc. of MKM 2003, Lecture Notes in Computer Science 2594, pp. 30–44, 2003.
- [Urb01] J. Urban, *Basic facts about inaccessible and measurable cardinals*, Formalized Mathematics, 9(2), pp. 323–329, 2001.